From: W. Olin Sibert

To: MTB Distribution

Date: December 2, 1980

Subject: Include Files for CONFIG Cards

This MTB describes a new set of include files to describe the "cards" in the CONFIG deck. These include files are intended as a complete replacement for the existing include files, which are incomplete, inconsistent, and occasionally incorrect. Programs which use the old include files, or which declared the cards they use in the program without using any include file at all, should be updated as time permits.

These files are constructed to be relatively compatible with existing usage. Much existing usage will have to be edited to take advantage of them, but the editing is all straightforward. The existing include files replaced by these new ones include: software_config_cards, hardware_config_cards, prph_card, intk_card, chnl_card, mpc_card, prph_tap_card, prph_dsk_card, and prt_config.

The benefit of these new include files is the consistency in naming conventions. These conventions can be seen in the text of the include files at the end of the MTB, and are also described in the next section.

Please send any comments, questions, etc., to the author:

By Multics Mail, at either MIT-Multics or System-M:

Sibert.Multics

By Honeywell Express Delivery Interoffice Mail:

W. Olin Sibert
Cambridge Information Systems Laboratory
HED MA22

(or) 575 Technology Square
Cambridge, Massachusetts 02139

Conventions:

All the new include files follow specific conventions, with the intent of making the files easier to use and the names easier to remember. For details on how these are implemented, refer to the include files themselves, which can be found at the end of this document. These conventions are:

* The include file for a card called NAME is called NAME_card.incl.pl1. The structure it declares is called NAME_card, and it is based on a pointer called NAME_cardp.

* Each structure begins with a char (4) value called NAME_card.word, which is the name of the card. Each include file also declares a char (4) constant, called NAME_CARD_WORD, (in this declaration ONLY, NAME is uppercase; all other uses of NAME are to be replaced by the lowercase name) which is the value in the word. It should be used for calls to config$find.

* Those fields which occur on several different config cards are named by the same name in all the include files. These names include: "iom" -- the IOM tag letter number for a peripheral channel, "chan" -- the IOM channel number for a peripheral, "nchan" -- the number of channels assigned to a peripheral, "port" -- the system controller port something is attached to, "tag" -- the value set in switches to identify a component, "drive" -- the index of a tape or disk drive, "subsystem" -- the name of a tape or disk subsystem, "ndrives" -- the number of tape or disk drives in a group, "state" -- the current state of a device: "on", "off", or possibly something else.

* For PRPH cards, the naming conventions are slightly different. In addition to prph_card.incl.pl1, there are include files for each known type of peripheral, called prph_TYPE_card.incl.pl1. The structures are called prph_TYPE_card, and based on prph_TYPE_cardp, as would be expected. There is no equivalent to the NAME_CARD_WORD constant, however.

* Each card ends with a declaration for type_word, which contains the two bit type codes for each field (NAME_card.field_type), and the field count (NAME_card.n_fields). The type codes are defined in config_deck.incl.pl1

* For cards which include an array of objects (such as device groups on a PRPH DSKx card), an additional structure is declared, called NAME_card_array, based on the same pointer as NAME_card. This declaration declares the variable size array with a calculation based on the contents of the n_fields portion of the card, and can be used with the hbound builtin to determine the number of elements in the variable size array. This is much preferable to doing arithmetic on n_fields directly, or looping until -1 is encountered, since it localizes the calculation to a single program (the include file) and makes it more robust in the face of format or size changes.

The following cards and include files are represented here:

----	config_deck.incl.pl1
BULK	bulk_card.incl.pl1
CHNL	chnl_card.incl.pl1
CLOK	clok_card.incl.pl1
CPU	cpu_card.incl.pl1
FNP	fnp_card.incl.pl1
INTK	intk_card.incl.pl1
IOM	iom_card.incl.pl1
MEM	mem_card.incl.pl1
MPC	mpc_card.incl.pl1
PAGE	page_card.incl.pl1
PARM	parm_card.incl.pl1
PART	part_card.incl.pl1
PRPH	prph_card.incl.pl1
PRPH CCUx	prph_ccu_card.incl.pl1
PRPH DSKx	prph_dsk_card.incl.pl1
PRPH OPC	prph_opc_card.incl.pl1
PRPH PRTx	prph_prt_card.incl.pl1
PRPH PUNx	prph_pun_card.incl.pl1
PRPH RDRx	prph_rdr_card.incl.pl1
PRPH TAPx	prph_tap_card.incl.pl1
ROOT	root_card.incl.pl1
SALV	salv_card.incl.pl1
SCHD	schd_card.incl.pl1
SST	sst_card.incl.pl1
TBLS	tbls_card.incl.pl1
TCD	tcd_card.incl.pl1
UDSK	udsk_card.incl.pl1

```
/* BEGIN INCLUDE FILE ... config_deck.incl.pl1 ... 11/13/80, W. Olin Sibert */

dcl (configp, cardp) pointer;
dcl  config_n_cards fixed bin;                          /* Number of cards used in config */
dcl  config_max_cards fixed bin;                        /* Max number of cards in config */

dcl  config_deck$ fixed bin external static;

dcl 1 config_deck aligned based (configp),
    2 cards (config_n_cards) aligned like config_card,
    2 pad_cards (config_max_cards - config_n_cards) aligned like config_card;

dcl 1 config_card aligned based (cardp),
    2 name char (4) aligned,
    2 data_field (14) bit (36) aligned,
    2 type_word aligned like config_card_type_word;

dcl 1 config_card_type_word aligned based,
    2 field_type (14) bit (2) unaligned,
    2 pad1 bit (4) unaligned,
    2 n_fields fixed bin (4) unsigned unaligned;

dcl (CONFIG_DECIMAL_TYPE      init ("11"b),
     CONFIG_OCTAL_TYPE        init ("00"b),
     CONFIG_SINGLE_CHAR_TYPE  init ("01"b),
     CONFIG_STRING_TYPE       init ("10"b)) bit (2) aligned static options (constant);

dcl  ZERO_CARD_NAME char (4) aligned internal static options (constant) init ("");
dcl  FREE_CARD_NAME char (4) aligned internal static options (constant) init ("");

dcl  VALID_CARD_NAME_CHARACTERS char (38) internal static options (constant) init
    ("abcdefghijklmnopqrstuvwxyz0123456789_.");          /* lowercase letters, digits, period and underscore */

/* END INCLUDE FILE config_deck.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... bulk_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  bulk_cardp pointer;                                /* pointer to BULK card */

dcl 1 bulk_card aligned based (bulk_cardp),             /* BULK card declaration */
    2 word char (4),                                    /* "BULK" */
    2 frec fixed bin,                                   /* First record used */
    2 nrec fixed bin,                                   /* Number of records used */
    2 port fixed bin (3),                               /* Controller port to which bulk store is attached */
    2 int_cell fixed bin (5),                           /* Interrupt Cell number */

    2 pad (10) bit (36) aligned,                        /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl  BULK_CARD_WORD char (4) aligned internal static options (constant) init ("bulk");

/* END INCLUDE FILE ... bulk_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... chnl_card.incl.pl1 ... October 1979, Michael R. Jordan */
/* Modified 11/11/80 W. Olin Sibert to add field_type, n_fields */

dcl  chnl_cardp ptr;                                    /* Pointer to a CHNL card. */

dcl 1 chnl_card based (chnl_cardp) aligned,             /* CHNL card declaration */
    2 word char (4),                                    /* "CHNL" */
    2 name char (4),                                    /* subsystem name */
    2 group (3),                                        /* Channel groups for this subsystem -- 9 fields total */
      3 iom fixed bin (3),                              /* IOM number */
      3 chan fixed bin (6),                             /* channel number */
      3 nchan fixed bin,                                /* number of channels */

    2 pad (4) bit (36) aligned,                         /* pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl 1 chnl_card_array aligned based (chnl_cardp),       /* Overlay for channel group array */
    2 pad1 (2) bit (36) aligned,
    2 group (min (3, divide (max (0, (chnl_card.n_fields - 1)), 3, 17, 0))),
      3 iom fixed bin (3),                              /* IOM number */
      3 chan fixed bin (6),                             /* Channel number. */
      3 nchan fixed bin;                                /* Number of logical channels on this channel. */

dcl  CHNL_CARD_WORD char (4) aligned internal static options (constant) init ("chnl");

/* END INCLUDE FILE ... chnl_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... clok_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  clok_cardp pointer;                                /* pointer to CLOK card */

dcl 1 clok_card aligned based (clok_cardp),             /* CLOK card declaration */
    2 word char (4),                                    /* "CLOK" */
    2 delta fixed bin,                                  /* Signed offset from GMT */
    2 zone char (4),                                    /* Name of time zone */
    2 boot_delta fixed bin,                             /* Number of hours allowed between bootloads */

    2 pad (11) bit (36) aligned,                        /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl  CLOK_CARD_WORD char (4) aligned internal static options (constant) init ("clok");

/* END INCLUDE FILE ... clok_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... cpu_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  cpu_cardp pointer;                                  /* pointer to CPU card */

dcl 1 cpu_card aligned based (cpu_cardp),                /* CPU card declaration */
    2 word char (4),                                     /* "CPU" */
    2 tag fixed bin (3),                                 /* CPU tag from switches, plus one */
    2 port fixed bin (3),                                /* Controller port for CPU */
    2 state char (4),                                    /* "ON", "OFF", "SHUT", or "TEST" */
    2 expander_port fixed bin (3),                       /* If present, indicates expander sub-port */

    2 pad (10) bit (36) aligned,                         /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,               /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;       /* number of fields used on card */

dcl  CPU_CARD_WORD char (4) aligned internal static options (constant) init ("cpu");

/* END INCLUDE FILE ... cpu_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... fnp_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  fnp_cardp pointer;                                  /* pointer to FNP card */

dcl 1 fnp_card aligned based (fnp_cardp),                /* FNP card declaration */
    2 word char (4),                                     /* "FNP" */
    2 tag fixed bin (3),                                 /* One more than FNP number in switches */
    2 iom fixed bin (3),                                 /* IOM to which it is attached */
    2 chan fixed bin (8),                                /* Channel number on IOM */

    2 pad (11) bit (36) aligned,                         /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,               /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;       /* number of fields used on card */

dcl  FNP_CARD_WORD char (4) aligned internal static options (constant) init ("fnp");

/* END INCLUDE FILE ... fnp_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... intk_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  intk_cardp pointer;                                    /* pointer to INTK card */

dcl 1 intk_card aligned based (intk_cardp),                 /* INTK card declaration */
    2 word char (4),                                        /* "INTK" */
    2 warm_or_cold char (4),                                /* Type of bootload: "WARM" or "COLD" */
    2 boot_drive fixed bin,                                 /* Tape drive on which MST is mounted */
    2 parms (12) char (4),                                  /* up to 12 arbitrary bootload parameters */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl 1 intk_card_array aligned based (intk_cardp),           /* Overlay for counting parameters */
    2 pad (3) bit (36) aligned,
    2 parms (max (0, intk_card.n_fields - 2)) bit (36) aligned;

dcl  INTK_CARD_WORD char (4) aligned internal static options (constant) init ("intk");

/* END INCLUDE FILE ... intk_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... iom_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  iom_cardp pointer;                                  /* pointer to IOM card */

dcl 1 iom_card aligned based (iom_cardp),                /* IOM card declaration */
    2 word char (4),                                     /* "IOM" */
    2 tag fixed bin (3),                                 /* One more than IOM tag set in maintenance panel switches */
    2 port fixed bin (3),                                /* Controller port to which IOM is connected */
    2 model char (4),                                    /* IOM model number: "6K" or "6KB" */
    2 state char (4),                                    /* State: "ON" or "OFF" */

    2 pad (10) bit (36) aligned,                         /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,               /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;       /* number of fields used on card */

dcl  IOM_CARD_WORD char (4) aligned internal static options (constant) init ("iom");

/* END INCLUDE FILE ... iom_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... mem_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  mem_cardp pointer;                                 /* pointer to MEM card */

dcl 1 mem_card aligned based (mem_cardp),               /* MEM card declaration */
    2 word char (4),                                    /* "MEM" */
    2 tag fixed bin (3),                                /* One more than module port to which controller is attached */
    2 size fixed bin (18),                              /* Number of pages in memory controller */
    2 state char (4),                                   /* State: "ON" or "OFF" */

    2 pad (11) bit (36) aligned,                        /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl  MEM_CARD_WORD char (4) aligned internal static options (constant) init ("mem");

/* END INCLUDE FILE ... mem_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... mpc_card.incl.pl1 ... June 1979, Michael R. Jordan */
/* Modified 11/11/80 W. Olin Sibert to add field_type, n_fields */

dcl  mpc_cardp ptr;                                     /* Pointer to an MPC card. */

dcl 1 mpc_card aligned based (mpc_cardp),
    2 word char (4),                                    /* Should be "mpc" */
    2 name char (4),                                    /* Name of this MPC - e.g., MSPA */
    2 model fixed bin,                                  /* Model of this MPC - e.g., 601. */
    2 port (4),                                         /* Per port information. 12 fields total */
      3 iom fixed bin (3),                              /* IOM number */
      3 chan fixed bin (6),                             /* Channel number. */
      3 nchan fixed bin,                                /* Number of logical channels on this channel. */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl 1 mpc_card_array aligned based (mpc_cardp),         /* Overlay for MPC port array */
    2 pad1 (3) bit (36) aligned,
    2 port (divide (max (0, (mpc_card.n_fields - 2)), 3, 17, 0)),
      3 iom fixed bin (3),                              /* IOM number */
      3 chan fixed bin (6),                             /* Channel number. */
      3 nchan fixed bin;                                /* Number of logical channels on this channel. */

dcl  MPC_CARD_WORD char (4) aligned internal static options (constant) init ("mpc");

/* END INCLUDE FILE ... mpc_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... page_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  page_cardp pointer;                                /* pointer to PAGE card */

dcl 1 page_card aligned based (page_cardp),             /* PAGE card declaration */
    2 word char (4),                                    /* "PAGE" */
    2 pd_name char (4),                                 /* Name of paging device */
    2 frec fixed bin,                                   /* First record to use */
    2 nrec fixed bin,                                   /* Number of records to use */

    2 del_pair (5),                                     /* Array listing deleted PD records */
      3 frec fixed bin,                                 /* First record in deleted group */
      3 nrec fixed bin,                                 /* Size of deleted group */

    2 pad (1) bit (36) aligned,                         /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl 1 page_card_array aligned based (page_cardp),       /* Overlay for counting deleted pairs */
    2 pad (4) bit (36) aligned,
    2 del_pair (divide (max (0, page_card.n_fields - 3), 2, 17, 0)),
      3 frec fixed bin,                                 /* First record in deleted group */
      3 nrec fixed bin;                                 /* Size of deleted group */

dcl  PAGE_CARD_WORD char (4) aligned internal static options (constant) init ("page");

/* END INCLUDE FILE ... page_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... parm_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl parm_cardp pointer;                                 /* pointer to PARM card */

dcl 1 parm_card aligned based (parm_cardp),             /* PARM card declaration */
    2 word char (4),                                    /* "PARM" */
    2 options (14) char (4),                            /* Parameters and their values */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl 1 parm_card_array based (parm_cardp),               /* Overlay for counting options */
    2 pad bit (36) aligned,
    2 options (parm_card.n_fields) bit (36) aligned;


dcl  parm_ptr pointer;                                  /* For use with config$find_parm */

dcl 1 numeric_parm aligned based (parm_ptr),            /* Overlay into middle of card for looking */
    2 name char (4),                                    /* at a parameter found by config$find_parm */
    2 value fixed bin (35);

dcl 1 string_parm aligned based (parm_ptr),
    2 name char (4),
    2 value char (4);

dcl  PARM_CARD_WORD char (4) aligned internal static options (constant) init ("parm");

/* END INCLUDE FILE ... parm_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... part_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  part_cardp pointer;                                    /* pointer to PART card */

dcl 1 part_card aligned based (part_cardp),                 /* PART card declaration */
    2 word char (4),                                        /* "PART" */
    2 name char (4),                                        /* Name of partition */
    2 subsystem char (4),                                   /* Disk subsystem name */
    2 drive fixed bin,                                      /* Drive number */
    2 real_name char (4),                          /* Real name of partition on volume (optional) */

    2 pad (10) bit (36) aligned,                            /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl 1 cold_part_card aligned based (part_cardp),            /* PART card declaration for cold boot */
    2 word char (4),                                        /* "PART" */
    2 name char (4),                                        /* Name of partition */
    2 subsystem char (4),                                   /* Disk subsystem name */
    2 drive fixed bin,                                      /* Drive number */
    2 highlow char (4),                                     /* Where to put it: "HIGH" or "LOW" */
    2 nrec fixed bin,                                       /* Number of records to be allocated */

    2 pad (9) bit (36) aligned,                             /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl  PART_CARD_WORD char (4) aligned internal static options (constant) init ("part");

/* END INCLUDE FILE ... part_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_card.incl.pl1 ... October 1979, Michael R. Jordan */
/* Modified 11/11/80 W. Olin Sibert to add field_type, n_fields */

dcl  prph_cardp ptr;                                    /* Pointer to a PRPH card. */

dcl 1 prph_card based (prph_cardp) aligned,             /* PRPH card declaration */
    2 word char (4),                                    /* "PRPH" */
    2 name char (4),                                    /* subsystem name */
    2 iom fixed bin (3),                                /* IOM number */
    2 chan fixed bin (6),                               /* channel number */
    2 model fixed bin,                                  /* model number */

    2 pad (10) bit (36) aligned,                        /* pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl  PRPH_CARD_WORD char (4) aligned internal static options (constant) init ("prph");

/* END INCLUDE FILE ... prph_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_ccu_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  prph_ccu_cardp pointer;                        /* pointer to PRPH card for Combination Card Unit */

dcl 1 prph_ccu_card aligned based (prph_ccu_cardp),  /* PRPH_CCU card declaration */
    2 word char (4),                                /* "PRPH" */
    2 name char (4),                                /* "CCUx" */
    2 iom fixed bin (2),                            /* IOM number */
    2 chan fixed bin (8),                           /* Channel number */
    2 model fixed bin,                              /* Model number of card punch */

    2 pad (10) bit (36) aligned,                    /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,          /* type of each field: see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;  /* number of fields used on card */

/* END INCLUDE FILE ... prph_ccu_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_dsk_card.incl.pl1 ... October 1979, Michael R. Jordan */
/* Modified 11/11/80 W. Olin Sibert to add field_type, n_fields */

dcl  prph_dsk_cardp ptr;                                    /* Pointer to a PRPH DSKx card. */

dcl 1 prph_dsk_card based (prph_dsk_cardp) aligned,         /* PRPH DSKx card declaration */
    2 word char (4),                                        /* "PRPH" */
    2 name char (4),                                        /* "DSKx" */
    2 iom fixed bin (3),                                    /* IOM number */
    2 chan fixed bin (6),                                   /* channel number */
    2 nchan fixed bin,                                      /* number of channels */

    2 group (5),
      3 model fixed bin,                                    /* model number */
      3 ndrives fixed bin,                                  /* number of drives */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl 1 prph_dsk_card_array aligned based (prph_dsk_cardp),   /* Overlay for drive group array */
    2 pad1 (5) bit (36) aligned,
    2 group (divide (max (0, (prph_dsk_card.n_fields - 4)), 2, 17, 0)),
      3 model fixed bin,                                    /* model number */
      3 ndrives fixed bin;                                  /* number of drives */

/* END INCLUDE FILE ... prph_dsk_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_opc_card.incl.pl1 ... 11/27/80, W. Olin Sibert */

dcl  prph_opc_cardp ptr;                                    /* pointer to PRPH OPC card */

dcl 1 prph_opc_card based (prph_opc_cardp) aligned,         /* PRPH OPC card declaration */
    2 word char (4),                                        /* "PRPH" */
    2 name char (4),                                        /* "OPC" */
    2 iom fixed bin (2),                                    /* IOM number */
    2 chan fixed bin (6),                                   /* channel number */
    2 model char (4),                                       /* console model number */

    2 buf_split fixed bin,                                  /* Percentage of buffer for syserr */
    2 pad (9) bit (36) aligned,                             /* pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

/* END INCLUDE FILE ... prph_opc_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_prt_card.incl.pl1 ... 11/11/80, W. Olin Sibert */

dcl  prph_prt_cardp ptr;                                    /* pointer to PRPH PRTx card */

dcl 1 prph_prt_card based (prph_prt_cardp) aligned,         /* PRPH PRTx card declaration */
    2 word char (4),                                        /* "PRPH" */
    2 devname char (4),                                     /* "PRTx" */
    2 iom fixed bin (2),                                    /* IOM number */
    2 chan fixed bin (6),                                   /* channel number */
    2 model fixed bin,                                      /* printer model number */

    2 train fixed bin,                                      /* print train ID */
    2 line_length fixed bin,                                /* printer line length */

    2 pad (8) bit (36) aligned,                             /* pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

/* END INCLUDE FILE ... prph_prt_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_pun_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  prph_pun_cardp pointer;                                    /* pointer to PRPH card for card punch */

dcl 1 prph_pun_card aligned based (prph_pun_cardp),             /* PRPH PUNx card declaration */
    2 word char (4),                                            /* "PRPH" */
    2 name char (4),                                            /* "PUNx" */
    2 iom fixed bin (2),                                        /* IOM number */
    2 chan fixed bin (8),                                       /* Channel number */
    2 model fixed bin,                                          /* Model number of card punch */

    2 pad (10) bit (36) aligned,                                /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                      /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;              /* number of fields used on card */

/* END INCLUDE FILE ... prph_pun_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_rdr_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  prph_rdr_cardp pointer;                        /* pointer to PRPH card for card reader */

dcl 1 prph_rdr_card aligned based (prph_rdr_cardp),  /* PRPH RDRx card declaration */
    2 word char (4),                                 /* "PRPH" */
    2 name char (4),                                 /* "PUNx" */
    2 iom fixed bin (2),                             /* IOM number */
    2 chan fixed bin (8),                            /* Channel number */
    2 model fixed bin,                               /* Model number of card punch */

    2 pad (10) bit (36) aligned,                     /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,           /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;   /* number of fields used on card */

/* END INCLUDE FILE ... prph_rdr_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... prph_tap_card.incl.pl1 ... October 1979, Michael R. Jordan */
/* Modified 11/11/80 W. Olin Sibert to add field_type, n_fields */

dcl  prph_tap_cardp ptr;                                    /* Pointer to PRPH TAPx card. */

dcl 1 prph_tap_card based (prph_tap_cardp) aligned,         /* PRPH TAPx card declaration */
    2 word char (4),                                        /* "PRPH" */
    2 name char (4),                                        /* "TAPx" */
    2 iom fixed bin (3),                                    /* IOM number */
    2 chan fixed bin (6),                                   /* channel number */
    2 model fixed bin,                                      /* model number */

    2 nchan fixed bin,                                      /* number of channels */
    2 nsysdrives fixed bin,                                 /* number of handlers reserved for system */
    2 max_concurrent fixed bin,                             /* max. handlers a user may attach */
    2 first9drive fixed bin (5),                            /* first 9-track handler number */
    2 n9drives fixed bin,                                   /* number of 9-track handlers */

    2 first7drive fixed bin (5),                            /* first 7-track handler number */
    2 n7drives fixed bin,                                   /* number of 7-track handlers */
    2 pad (3) bit (36) aligned,                             /* pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

/* END INCLUDE FILE ... prph_tap_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... root_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  root_cardp pointer;                                /* pointer to ROOT card */

dcl 1 root_card aligned based (root_cardp),             /* ROOT card declaration */
    2 word char (4),                                    /* "ROOT" */
    2 volume (7),                                       /* List of RLV volumes. RPV is first in the list */
      3 subsystem char (4),                             /* Disk subsystem name */
      3 drive fixed bin,                                /* Disk drive number */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl 1 root_card_array aligned based (root_cardp),       /* Overlay for counting volumes */
    2 pad bit (36) aligned,
    2 volume (divide (root_card.n_fields, 2, 17, 0)),
      3 subsystem char (4),                             /* Disk subsystem name */
      3 drive fixed bin;                                /* Disk drive number */

dcl  ROOT_CARD_WORD char (4) aligned internal static options (constant) init ("root");

/* END INCLUDE FILE ... root_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... salv_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl salv_cardp pointer;                                 /* pointer to SALV card */

dcl 1 salv_card aligned based (salv_cardp),             /* SALV card declaration */
    2 word char (4),                                    /* "SALV" */
    2 options (14) char (4),                            /* Options for salvaging with */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl 1 salv_card_array based (salv_cardp),               /* Overlay for counting options */
    2 pad bit (36) aligned,
    2 options (salv_card.n_fields) bit (36) aligned;

dcl  SALV_CARD_WORD char (4) aligned internal static options (constant) init ("salv");

/* END INCLUDE FILE ... salv_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... schd_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  schd_cardp pointer;                                    /* pointer to SCHD card */

dcl 1 schd_card aligned based (schd_cardp),                 /* SCHD card declaration */
    2 word char (4),                                        /* "SCHD" */
    2 ws_factor fixed bin (35, 18),                         /* Working Set Factor */
    2 tefirst fixed bin,                                    /* tefirst (in 1/8 second units) */
    2 telast fixed bin,                                     /* telast (in 1/8 second units) */
    2 timax fixed bin,                                      /* timax (in 1/8 second units) */

    2 min_eligible fixed bin,                               /* minimum number of eligible processes */
    2 max_eligible fixed bin,                               /* maximum number of eligible processes */
    2 max_max_eligible fixed bin,                           /* upper limit on max_eligible -- # of stack_0 segments */
    2 post_purging char (4),                                /* Whether to post-purge: "ON" or "OFF" */

    2 pad (6) bit (36) aligned,                             /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl  SCHD_CARD_WORD char (4) aligned internal static options (constant) init ("schd");

/* END INCLUDE FILE ... schd_card.incl.pl1 */
```

MTB 480: Include Files For CONFIG Cards

```
/* BEGIN INCLUDE FILE ... sst_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  sst_cardp pointer;                                 /* pointer to SST card */

dcl 1 sst_card aligned based (sst_cardp),               /* SST card declaration */
    2 word char (4),                                    /* "SST" */
    2 no_aste (0:3) fixed bin,                          /* Size of the four ASTE pools */

    2 pad (10) bit (36) aligned,                        /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl  SST_CARD_WORD char (4) aligned internal static options (constant) init ("sst");

/* END INCLUDE FILE ... sst_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... tbls_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  tbls_cardp pointer;                                    /* pointer to TBLS card */

dcl 1 tbls_card aligned based (tbls_cardp),                 /* TBLS card declaration */
    2 word char (4),                                        /* "TBLS" */
    2 table (7),                                            /* Array of table names and sizes */
      3 name char (4),                                      /* Name of table */
      3 size fixed bin,                                     /* Size of table */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl 1 tbls_card_array aligned based (tbls_cardp),           /* Overlay for counting tables */
    2 pad bit (36) aligned,
    2 table (divide (tbls_card.n_fields, 2, 17, 0)),
      3 name char (4),                                      /* Name of table */
      3 size fixed bin;                                     /* Size of table */

dcl  TBLS_CARD_WORD char (4) aligned internal static options (constant) init ("tbls");

/* END INCLUDE FILE ... tbls_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... tcd_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  tcd_cardp pointer;                                 /* pointer to TCD card */

dcl 1 tcd_card aligned based (tcd_cardp),               /* TCD card declaration */
    2 word char (4),                                    /* "TCD" */
    2 no_apt fixed bin,                                 /* Number of APT entries */
    2 no_itt fixed bin,                                 /* Number of ITT entries */
    2 no_dst fixed bin,                                 /* Number of DST entries */
    2 max_hproc_segno fixed bin,                        /* Optional max segno for collection 2 */

    2 pad (10) bit (36) aligned,                        /* Pad to 15 fields */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,              /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;      /* number of fields used on card */

dcl  TCD_CARD_WORD char (4) aligned internal static options (constant) init ("tcd");

/* END INCLUDE FILE ... tcd_card.incl.pl1 */
```

```
/* BEGIN INCLUDE FILE ... udsk_card.incl.pl1 ... 11/27/80 W. Olin Sibert */

dcl  udsk_cardp pointer;                                    /* pointer to UDSK card */

dcl 1 udsk_card aligned based (udsk_cardp),                 /* UDSK card declaration */
    2 word char (4),                                        /* "UDSK" */
    2 subsystem char (4),                                   /* Name of disk subsystem */
    2 nchan fixed bin,                                      /* Max number of channels usable by ioi */

    2 group (6),                                            /* Groups of drives available for ioi use */
      3 drive fixed bin,                                    /* Index of first drive in group */
      3 ndrives fixed bin,                                  /* Number of drives in group */

    2 type_word aligned,
      3 field_type (14) bit (2) unaligned,                  /* type of each field; see config_deck.incl.pl1 */
      3 pad1 bit (4) unaligned,
      3 n_fields fixed bin (4) unsigned unaligned;          /* number of fields used on card */

dcl 1 udsk_card_array aligned based (udsk_cardp),           /* Overlay for counting drive groups */
    2 pad (3) bit (36) aligned,
    2 group (divide (max (0, udsk_card.n_fields - 2), 2, 17, 0)),
      3 drive fixed bin,                                    /* Index of first drive in group */
      3 ndrives fixed bin;                                  /* Number of drives in group */

dcl  UDSK_CARD_WORD char (4) aligned internal static options (constant) init ("udsk");

/* END INCLUDE FILE ... udsk_card.incl.pl1 */
```